
	Reason for change:
	Clause 7.2A.5 extends the P-Charging-Vector header field.

However, the syntax in table 7.2A.5 lists the extension-param three times, two of them also having different values.

[…]
   extension-param = token [EQUAL token]
[…]
   extension-param = token [EQUAL (token | quoted-string)]
[…]
   extension-param = token [EQUAL (token | quoted-string)]
[…]

It is therefore proposed to delete two of these entries.

1st change
[bookmark: _Toc509938814]7.2A.5.2.1	General
The syntax of the P-Charging-Vector header field is described in RFC 7315 [52]. There may be additional coding rules for this header field depending on the type of IP-CAN, according to access technology specific descriptions.
Table 7.2A.5 describes 3GPP-specific extensions to the P-Charging-Vector header field defined in RFC 7315 [52].
Table 7.2A.5: Syntax of extensions to P-Charging-Vector header field

   access-network-charging-info = (gprs-charging-info / i-wlan-charging-info / xdsl-charging-info / packetcable-charging-info / icn-charging-info / eps-charging-info / eth-charging-info/ loopback-indication / generic-param)
   gprs-charging-info = ggsn SEMI auth-token [SEMI pdp-info-hierarchy] *(SEMI extension-param)
   ggsn = "ggsn" EQUAL gen-value
   pdp-info-hierarchy = "pdp-info" EQUAL LDQUOT pdp-info *(COMMA pdp-info) RDQUOT
   pdp-info = pdp-item SEMI pdp-sig SEMI gcid [SEMI flow-id]
   pdp-item = "pdp-item" EQUAL DIGIT
   pdp-sig = "pdp-sig" EQUAL ("yes" / "no")
   gcid = "gcid" EQUAL 1*HEXDIG
   auth-token = "auth-token" EQUAL 1*HEXDIG
   flow-id = "flow-id" EQUAL "(" "{" 1*DIGIT COMMA 1*DIGIT "}" *(COMMA "{" 1*DIGIT COMMA 1*DIGIT "}")")"
   extension-param = token [EQUAL token]
   i-wlan-charging-info = "pdg"
   xdsl-charging-info = bras SEMI auth-token [SEMI xDSL-bearer-info] *(SEMI extension-param)
   bras = "bras" EQUAL gen-value
   xDSL-bearer-info = "dsl-bearer-info" EQUAL LDQUOT dsl-bearer-info *(COMMA dsl-bearer-info) RDQUOT
   dsl-bearer-info = dsl-bearer-item SEMI dsl-bearer-sig SEMI dslcid [SEMI flow-id]
   dsl-bearer-item = "dsl-bearer-item" EQUAL DIGIT
   dsl-bearer-sig = "dsl-bearer-sig" EQUAL ("yes" / "no")
   dslcid = "dslcid" EQUAL 1*HEXDIG
   packetcable-charging-info = packetcable [SEMI bcid]
   packetcable = "packetcable-multimedia"
   bcid = "bcid" EQUAL 1*48(HEXDIG)
   icn-charging-info = icn-bcp *(SEMI itid) [SEMI extension-param]
   icn-bcp = "icn-bcp" EQUAL gen-value
   itid = itc-sig SEMI itc-id SEMI *(flow-id)
   itc-sig = "itc-sig" EQUAL ("yes" / "no")
   itc-id = "itc-id" EQUAL gen-value
   flow-id = "flow-id" EQUAL gen-value
   extension-param = token [EQUAL (token | quoted-string)]
   eps-charging-info = pdngw [SEMI eps-bearer-hierarchy] *(SEMI extension-param)
   pdngw = "pdngw" EQUAL gen-value
   eps-bearer-hierarchy = "eps-info" EQUAL LDQUOT eps-info *(COMMA eps-info) RDQUOT
   eps-info = eps-item SEMI eps-sig SEMI ecid [SEMI flow-id]
   eps-item = "eps-item" EQUAL DIGIT
   eps-sig = "eps-sig" EQUAL ("yes" / "no")
   ecid = "ecid" EQUAL 1*HEXDIG
   eth-charging-info = ip-edge *(SEMI extension-param)
   fiber-charging-info = ip-edge *(SEMI extension-param)
   ip-edge = "ip-edge" EQUAL gen-value
   extension-param = token [EQUAL (token | quoted-string)]
   loopback-indication = "loopback"
   fe-identifier = "fe-identifier" EQUAL fe-id-list   fe-id-list = DQUOTE fe-id-param *(COMMA fe-id-param) DQUOTE
   fe-id-param = fe-addr/as-addr
   fe-addr = "fe-addr" EQUAL gen-value
   as-addr = "as-addr" EQUAL gen-value "-" ap-id
   ap-id = "ap-id" EQUAL gen-value


NOTE:	The syntax above is not aligned with the rules for defining new P-Charging-Vector header field parameters as defined in RFC 7315 [52]. Entities that perform syntax check (even if they are not interested in specific header field parameter values) of the header field need to follow the explicit syntax above, as using the rules in RFC 7315 [52] would trigger a parser error.
The access-network-charging-info parameter is an instance of generic-param from the current charge-params component of P-Charging-Vector header field.
The access-network-charging-info parameter includes alternative definitions for different types access networks. The description of these parameters are given in the subsequent subclauses.
The "access-network-charging-info" header field parameter is not included in the P-Charging-Vector for SIP signalling that is not associated with a session.
When the "access-network-charging-info" is included in the P-Charging-Vector and necessary information is not available from the IP-CAN (e.g. via Gx/Rx interface) reference points then null or zero values are included.
For type 1 and type 3 IOIs, the generating SIP entity shall express the "orig-ioi" and "term-ioi" header field parameters in the format of a quoted string as specified in RFC 7315 [52]. 
If an IOI is a type 1 IOI, the content of the quoted string consists of the "Type 1" string prefix followed by the IOI value. The "Type 1" string prefix is the type-1-prefix value specified in the table 7.2A.5A.
If an IOI is a type 3 IOI, the content of the quoted string consists of the "Type 3" string prefix followed by the IOI value. The "Type 3" string prefix is the type-3-prefix value specified in the table 7.2A.5A.
Table 7.2A.5A: String prefixes

type-1-prefix = %x54.79.70.65.20.31 ; "Type 1"
type-3-prefix = %x54.79.70.65.20.33 ; "Type 3"


If an IOI is a type 2 IOI, the value of the "orig-ioi" and "term-ioi" header field parameters is set to the IOI value. No string prefix is used.
The receiving SIP entity does not perform syntactic checking of the contents of the IOI parameter (the IOI parameter is passed unmodified to charging entities).
The "loopback" parameter is provided to the charging system of other entities in the signalling path to indicate that loopback has been applied and entities of the IM CN subsystem involved in the loopback, e.g. TRF, can have generated CDRs in their own right.
The "fe-identifier" header field parameter is an instance of generic-param from the current charge-params component of the P-Charging-Vector header field. This header field parameter contains one or more IM CN subsystem functional entity addresses ("fe-addr") and/or AS addresses ("as-addr") and application identifiers ("ap-id") where the IM CN subsystem functional entity does create charging information for the related CDR of this IM CN subsystem functional entity. For AS hosting several applications the AS address can appear several times, each accompanied with a different application identifier based on the application executed by the AS.
End of changes



